** Clearing Stata memory
capture log close
clear all
set more off, perm
set seed 1234

/////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////// Table 8: Average Annual Wages between 7 and 12 Years After Admission Exam by Gender /////////////////////////////////////////////////////////////////////////////  ////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////


** Opening Phase 2 norm_scores dataset 
use "Work Data/Gender_Phase2_long.dta",clear

*** Creating variables
encode subject, gen (sub)
tab subject, gen (d_sub)
label var sub "Subject"

** Subject dummies
rename d_sub1 Biology
rename d_sub2 Chemistry
rename d_sub3 Geography
rename d_sub4 History
rename d_sub5 Language
rename d_sub6 Mathematics
rename d_sub7 Physics
rename d_sub8 Portuguese
* Labels
label var Biology "Biology"
label var Chemistry "Chemistry"
label var Geography "Geography"
label var History "History"
label var Math "Mathematics"
label var Physics "Physics"
label var Portuguese "Portuguese"
label var Language "Foreign Language"

** Interaction: priority X female
gen fem_priority=female*priority
label var fem_priority "Female $\times$ Priority"

** Interaction: priority X female
foreach v of varlist Biology-Portuguese {
gen fem_`v'=`v'*female
label var fem_`v' "Female $\times$ `v'"
gen prio_`v'=priority*`v'
label var prio_`v' "Priority $\times$ `v'"
gen fem_prio_`v'=fem_priority*`v'
label var fem_prio_`v' "Female $\times$ Priority $\times$ `v'"
}

global subject "Chemistry Geography History Mathematics Physics"
global subject_fem "fem_Chemistry fem_Geography fem_History fem_Mathematics fem_Physics"

** P1 scores: P1 normalized subject-specific scores
forvalues i=2(1)4 {
gen norm_p1score`i'=norm_p1score^`i'
sum norm_p1score`i'
}

*********************************************************************************
****************   Relative performances ****************************************
*********************************************************************************

******************************** ENEM ********************************

foreach v in norm_enem_w {
bys year female: egen `v'_ave_g=mean(`v')
gen `v'_g=`v'-`v'_ave_g
bys year female: sum `v'_g
}
drop norm_enem_w_ave_g

* Priority x relative performance in ENEM:
foreach v in  norm_enem_w {
gen `v'_priority_g=`v'_g*priority
forvalues i=2(1)4 {
gen `v'_priority_g`i'=`v'_g^`i'*priority
sum `v'_priority_g`i'
}
}

global g_norm_enem_w_prio norm_enem_w_priority_g*
d $g_norm_enem_w_prio

* Interaction: subject X ENEM
foreach v of varlist Biology-Portuguese {
gen enem_`v'=`v'*norm_enem_w_g
label var enem_`v' "ENEM $\times$ `v'"
gen fem_enem_`v'=female*norm_enem_w_g*`v'
label var fem_enem_`v' "Female $\times$ ENEM $\times$ `v'"
forvalues i=2(1)4 {
gen enem_`v'_`i'=enem_`v'^`i'
gen fem_enem_`v'_`i'=fem_enem_`v'^`i'
}
sum enem_`v'* fem_enem_`v'*
}

global g_pol_enem_sub "enem_Chemistry* enem_Geography* enem_History* enem_Mathematics* enem_Physics*"
d $g_pol_enem_sub

******************************** Phase 1 scores ********************************

foreach v in norm_p1score  {
bys year subject female: egen gs_`v'_ave=mean(`v')
gen gs_`v'=`v'-gs_`v'_ave
bys year female subject: sum gs_`v'
drop gs_`v'_ave

forvalues i=2(1)4 {
gen gs_`v'`i'=gs_`v'^`i'
sum gs_`v'`i'
}

global gs_pol_`v' gs_`v' gs_`v'2 gs_`v'3 gs_`v'4
d $gs_pol_`v'

* Priority x Phase 1 scores:
gen gs_`v'_prio=gs_`v'*priority
forvalues i=2(1)4 {
gen gs_`v'_prio`i'=gs_`v'`i'*priority
sum gs_`v'_prio*
}
}

global gs_pol_norm_p1score_prio gs_norm_p1score_prio*
d $gs_pol_norm_p1score_prio

*********************************************************************************
**************** Main sample ****************************************************
*********************************************************************************

* 1) Only years before the affirmative action took place
drop if aa_year==1
tab year
drop if year==2000
tab year

* 2) Drop Portuguese and Foreign Language (in Phase 1 there is no Portuguese or Foreign Language exams - For Portuguese Phase 1 has an essay)
 tab subject, sum(norm_p1score)
 drop if subject=="lang" | subject=="port" 
 tab subject, sum(norm_p1score)
 drop Language Portuguese prio_Language prio_Portuguese fem_prio_Language fem_prio_Portuguese 
 
 ******************************** SAVE DATA SET TO BOOTSTRAP ********************************
 
keep norm_score fem_priority priority $subject $subject_fem $gs_pol_norm_p1score $g_norm_enem_w_prio $gs_pol_norm_p1score_prio inscri2 female ///
enem norm_enem_w gen_ques_st1 essay_st1 total_st1 career_choice year $g_pol_enem_sub

save "Work Data/Temp_Bootstrap.dta", replace

********************************************************************************************************
****************  Step 1: In the first step we obtain initial estimates and store them  ****************
********************************************************************************************************

** # 1 - Main specification (column 6)	
reghdfe norm_score priority $subject $subject_fem $g_pol_enem_sub $g_norm_enem_w_prio $gs_pol_norm_p1score $gs_pol_norm_p1score_prio, cluster(inscri2) absorb(inscri2) resid
** # 2 - Save residuals 
predict residuals, resid
** # 3 - Create relative priority performance
ttest residuals, by(female) // Residuals close to zero in both cases
ttest residuals if priority==1, by(female)
sum residuals if female==1 & priority==1

bys inscri2: egen res_prio_ave=mean(residuals) if priority==1
bys inscri2: egen res_prio_average=min(res_prio_ave)
mdesc res_prio_average
tab career_choice if  res_prio_average==.
bys inscri2: egen res_nonprio_ave=mean(residuals) if priority==0
bys inscri2: egen res_nonprio_average=min(res_nonprio_ave)
mdesc res_nonprio_average
gen res_diff=res_prio_average-res_nonprio_average
mdesc res_diff
ttest res_diff, by(female)

collapse (mean) res_diff female enem norm_enem_w gen_ques_st1 essay_st1 total_st1 career_choice /* expected_duration* */ year, by(inscri2)

count if res_diff==.

** # 3 - Check if the gender average differences in the residuals are similar to  Female x Priority main coefficient
ttest res_diff, by(female) // diff | .040619

 ***************************************************
 ****************** Wages RAIS *********************
 ***************************************************

count
merge 1:1 inscri2 using "Work Data/RAIS_cleaned.dta"
drop if _merge==2
tab _merge  

** # 4 - Wage measures: 

sum mwagetot* decwagetot*

forvalues i=7(1)12 {
gen yearafter`i'=year+`i'
tab yearafter`i', mi
gen annual_wage_after`i'=.
levelsof yearafter`i', local(levels) 
foreach l of local levels {
replace annual_wage_after`i'=mwagetot`l' if yearafter`i'==`l'
}
}

**** Average and maximum wage

foreach x in annual_wage {
    
* Average
egen avg_`x'_712=rowmean(`x'_after7 `x'_after8 `x'_after9 `x'_after10 `x'_after11 `x'_after12) // 7-12 years after admission exam
sum avg_`x'*
sum avg_`x'* if _merge==1

* Maximum
egen max_`x'_712=rowmax(`x'_after7 `x'_after8 `x'_after9 `x'_after10 `x'_after11 `x'_after12) // 7-12 years after admission exam
sum max_`x'*
sum max_`x'* if _merge==1

}

* Log variables
foreach v of varlist annual_wage_after*  max* avg*  {
 gen l_`v'=log(`v')
 sum l_`v'
}

* Normalize residual measure
foreach v of varlist res_diff {
sum `v'
gen mean_`v'=r(mean)
gen sd_`v'=r(sd)
gen norm_`v'=(`v'-mean_`v')/sd_`v'
sum  norm_`v'
}

******************************** SAVE WAGES DATA TO MERGE WITH DATA SET TO BOOTSTRAP ********************************
preserve
keep l_*_annual_wage_712 inscri2 res_diff norm_res_diff career_choice norm_enem_w female year /*predict_year_degree*/
renvarlab res_diff norm_res_diff career_choice norm_enem_w female, postfix(_orig)  
duplicates report
save "Work Data/Temp_Wages_Bootstrap.dta", replace
restore

** # 5 - Control for the residuals 
drop if res_diff==.

**********  Annual Wages and Participation Formal Labor Market **********

gen gender="Female" if female==1
replace gender="Male" if female==0

levelsof gender, local(levels) 

foreach g of local levels {

foreach v of varlist l_*_annual_wage_712 {

preserve

keep if gender=="`g'"
tab gender

capture erase "Work Data/BS`v'_`g'.dta"

** Do not control for program FE

reghdfe `v' norm_res_diff, vce(robust) absorb(year)
foreach x in _cons norm_res_diff {
gen b_`x'1=_b[`x']
gen se_`x'1=_se[`x']	
}

reghdfe `v' norm_enem_w, vce(robust) absorb(year)
foreach x in _cons norm_enem_w  {
gen b_`x'2=_b[`x']
gen se_`x'2=_se[`x']	
}

reghdfe `v' norm_res_diff norm_enem_w, vce(robust) absorb(year)
foreach x in _cons norm_res_diff norm_enem_w  {
gen b_`x'3=_b[`x']
gen se_`x'3=_se[`x']	
}

** Control for program FE

reghdfe `v' norm_res_diff, vce(robust) absorb(career_choice year)
foreach x in _cons norm_res_diff {
gen b_`x'4=_b[`x']
gen se_`x'4=_se[`x']	
}

reghdfe `v' norm_enem_w, vce(robust) absorb(career_choice year)
foreach x in _cons norm_enem_w  {
gen b_`x'5=_b[`x']
gen se_`x'5=_se[`x']	
}

reghdfe `v' norm_res_diff norm_enem_w, vce(robust) absorb(career_choice year)
foreach x in _cons norm_res_diff norm_enem_w  {
gen b_`x'6=_b[`x']
gen se_`x'6=_se[`x']	
}

gen boot_run=0
keep boot_run b_* se_*
keep if _n==1  
count
sum

save "Work Data/BS`v'_`g'.dta", replace

restore

}
}

********************************************************************************************************
**************************************  Step 2: Bootstrap   ********************************************
********************************************************************************************************

use "Work Data/Temp_Bootstrap.dta", clear
merge m:1 inscri2 using "Work Data/Temp_Wages_Bootstrap.dta"
drop _merge
count 

local boots = 999

gen gender="Female" if female==1
replace gender="Male" if female==0

levelsof gender, local(levels) 
foreach g of local levels {

quietly {
forvalues i=1(1)`boots' {

noisily display "Working on `i' out of `boots' at $S_TIME , gender `g'"

foreach v of varlist l_*_annual_wage_712 {

preserve

local seed=2023 + `i'
set seed `seed'

bsample,  cluster(inscri2) idcluster(newid)  

********** Estimating the residuals **********

** # 1 - Main specification (column 6)	
reghdfe norm_score priority $subject $subject_fem $g_pol_enem_sub $g_norm_enem_w_prio $gs_pol_norm_p1score $gs_pol_norm_p1score_prio, cluster(newid) absorb(newid) resid
** # 2 - Save residuals 
predict residuals, resid
** # 3 - Create relative priority performance
bys newid: egen res_prio_ave=mean(residuals) if priority==1
bys newid: egen res_prio_average=min(res_prio_ave)
mdesc res_prio_average
bys newid: egen res_nonprio_ave=mean(residuals) if priority==0
bys newid: egen res_nonprio_average=min(res_nonprio_ave)
mdesc res_nonprio_average
gen res_diff=res_prio_average-res_nonprio_average
mdesc res_diff
ttest res_diff, by(female) 

collapse (mean) res_diff female enem norm_enem_w gen_ques_st1 essay_st1 total_st1 year career_choice `v' (firstnm) gender, by(newid)

drop if res_diff==. 
count if res_diff==.
ttest res_diff, by(female)

* Normalizing residuals
foreach r of varlist res_diff {
sum `r'
gen mean_`r'=r(mean)
gen sd_`r'=r(sd)
gen norm_`r'=(`r'-mean_`r')/sd_`r'
sum  norm_`r'
}

keep if gender=="`g'"
tab gender

** Do not control for program FE

reghdfe `v' norm_res_diff, vce(robust) absorb(year)
foreach x in _cons norm_res_diff {
gen b_`x'1=_b[`x']
gen se_`x'1=_se[`x']	
}

reghdfe `v' norm_enem_w, vce(robust) absorb(year)
foreach x in _cons norm_enem_w  {
gen b_`x'2=_b[`x']
gen se_`x'2=_se[`x']	
}

reghdfe `v' norm_res_diff norm_enem_w, vce(robust) absorb(year)
foreach x in _cons norm_res_diff norm_enem_w  {
gen b_`x'3=_b[`x']
gen se_`x'3=_se[`x']	
}

** Control for program FE

reghdfe `v' norm_res_diff, vce(robust) absorb(career_choice year)
foreach x in _cons norm_res_diff {
gen b_`x'4=_b[`x']
gen se_`x'4=_se[`x']	
}

reghdfe `v' norm_enem_w, vce(robust) absorb(career_choice year)
foreach x in _cons norm_enem_w  {
gen b_`x'5=_b[`x']
gen se_`x'5=_se[`x']	
}

reghdfe `v' norm_res_diff norm_enem_w, vce(robust) absorb(career_choice year)
foreach x in _cons norm_res_diff norm_enem_w  {
gen b_`x'6=_b[`x']
gen se_`x'6=_se[`x']	
}

gen boot_run=`i'
keep boot_run b_*
keep if _n==1  
summarize

append using "Work Data/BS`v'_`g'.dta" 
save "Work Data/BS`v'_`g'.dta", replace

restore
}

}
}
}

********************************************************************************************************
*********************** Step 3: Generating the bootstrapped standard errors ****************************
********************************************************************************************************

matrix drop _all

use "Work Data/Temp_Wages_Bootstrap.dta", clear

gen gender="Female" if female==1
replace gender="Male" if female==0

levelsof gender, local(levels) 
foreach gender of local levels {

use "Work Data/Temp_Wages_Bootstrap.dta", clear

foreach y of varlist l_*_annual_wage_712 {
	
use "Work Data/BS`y'_`gender'.dta", clear

gen gender="`gender'"
tab gender

** Calculating bootstrapped standard errors

foreach x in norm_enem_w norm_res_diff _cons {
forvalues i = 1(1)6 {
capture egen BS_sd_`x'`i'=sd(b_`x'`i')  if boot_run~=0
}
}

if gender=="Female" local g = "F"
if gender=="Male" local  g = "M"

noisily display  "Gender `gender' `g'"

** Making a matrix with bootstrapped standard errors

matrix define bse`g'`y'1 = J(1,2,0)
matrix define bse`g'`y'2 = J(1,2,0)
matrix define bse`g'`y'3 = J(1,3,0)
matrix define bse`g'`y'4 = J(1,2,0)
matrix define bse`g'`y'5 = J(1,2,0)
matrix define bse`g'`y'6 = J(1,3,0)

matrix colnames bse`g'`y'1 = resid _cons
matrix colnames bse`g'`y'2 = norm_enem_w _cons
matrix colnames bse`g'`y'3 = resid norm_enem_w _cons
matrix colnames bse`g'`y'4 = resid _cons
matrix colnames bse`g'`y'5 = norm_enem_w _cons
matrix colnames bse`g'`y'6 = resid norm_enem_w _cons

foreach i in 1 3 4 6 {
sum BS_sd_norm_res_diff`i'
scalar BS_sd_norm_res_diff = r(mean)
matrix bse`g'`y'`i'[1,1] = BS_sd_norm_res_diff
}

foreach i in 2 5 {
sum BS_sd_norm_enem_w`i'
scalar BS_sd_norm_enem_w = r(mean)
matrix bse`g'`y'`i'[1,1] = BS_sd_norm_enem_w
}

foreach i in 3 6 {
sum BS_sd_norm_enem_w`i'
scalar BS_sd_norm_enem_w = r(mean)
matrix bse`g'`y'`i'[1,2] = BS_sd_norm_enem_w
}

foreach i in 1 2 4 5 {
sum BS_sd__cons`i'
scalar BS_sd_cons = r(mean)
matrix bse`g'`y'`i'[1,2] = BS_sd_cons
}

foreach i in 3 6 {
sum BS_sd__cons`i'
scalar BS_sd_cons = r(mean)
matrix bse`g'`y'`i'[1,3] = BS_sd_cons
}

forvalues i = 1(1)6 {
matrix list bse`g'`y'`i'
}
}

}

********************************************************************************************************
*********************** Step 4: Computing p-values and exporting tables ********************************
********************************************************************************************************

use "Work Data/Temp_Wages_Bootstrap.dta", clear
rename norm_res_diff_orig resid
rename career_choice_orig career_choice
rename norm_enem_w_orig norm_enem_w
rename female_orig female

label var norm_enem_w "Norm. ENEM scores"
label var female "Female"
label var resid "Relative priority performance"
 
gen gender_short="F" if female==1
replace gender_short="M" if female==0

levelsof gender_short, local(levels) 
foreach g of local levels {
	
preserve

keep if gender_short=="`g'"
tab gender_short

**********  Annual Wages **********

drop if resid==.

foreach y of varlist l_*_annual_wage_712 {

** Do not control for program FE

reghdfe `y' resid, vce(robust) absorb(year)
matrix define b`g'`y'1 = e(b)
scalar DF`g'`y'1=e(df_r)
estimates store `g'`y'1
estadd ysumm
* Adding bootstrapped standard errors to the estimation results.
estimates restore `g'`y'1
estadd matrix seboot =  bse`g'`y'1

reghdfe `y' norm_enem_w, vce(robust) absorb(year)
matrix define b`g'`y'2 = e(b)
scalar DF`g'`y'2=e(df_r)
estimates store `g'`y'2
estadd ysumm
* Adding bootstrapped standard errors to the estimation results.
estimates restore `g'`y'2
estadd matrix seboot =  bse`g'`y'2

reghdfe `y' resid norm_enem_w, vce(robust) absorb(year)
matrix define b`g'`y'3 = e(b)
scalar DF`g'`y'3=e(df_r)
estimates store `g'`y'3
estadd ysumm
* Adding bootstrapped standard errors to the estimation results.
estimates restore `g'`y'3
estadd matrix seboot =  bse`g'`y'3

** Control for program FE

reghdfe `y' resid, vce(robust) absorb(career_choice year)
matrix define b`g'`y'4 = e(b)
scalar DF`g'`y'4=e(df_r)
estimates store `g'`y'4
estadd ysumm
* Adding bootstrapped standard errors to the estimation results.
estimates restore `g'`y'4
estadd matrix seboot =  bse`g'`y'4

reghdfe `y' norm_enem_w, vce(robust) absorb(career_choice year)
matrix define b`g'`y'5 = e(b)
scalar DF`g'`y'5=e(df_r)
estimates store `g'`y'5
estadd ysumm
* Adding bootstrapped standard errors to the estimation results.
estimates restore `g'`y'5
estadd matrix seboot =  bse`g'`y'5

reghdfe `y' resid norm_enem_w, vce(robust) absorb(career_choice year)
matrix define b`g'`y'6 = e(b)
scalar DF`g'`y'6=e(df_r)
estimates store `g'`y'6
estadd ysumm
* Adding bootstrapped standard errors to the estimation results.
estimates restore  `g'`y'6
estadd matrix seboot =  bse`g'`y'6

estadd local program "Yes": `g'`y'4 `g'`y'5 `g'`y'6 
estadd local program "No": `g'`y'1 `g'`y'2 `g'`y'3  
estadd local year "Yes": `g'`y'*

* Calculating p-values

foreach x in pval pvalue {	
	
matrix define b`x'`g'`y'1 = J(1,2,0)
matrix define b`x'`g'`y'2 = J(1,2,0)
matrix define b`x'`g'`y'3 = J(1,3,0)
matrix define b`x'`g'`y'4 = J(1,2,0)
matrix define b`x'`g'`y'5 = J(1,2,0)
matrix define b`x'`g'`y'6 = J(1,3,0)

matrix colnames b`x'`g'`y'1 = resid _cons
matrix colnames b`x'`g'`y'2 = norm_enem_w _cons
matrix colnames b`x'`g'`y'3 = resid norm_enem_w _cons
matrix colnames b`x'`g'`y'4 = resid _cons
matrix colnames b`x'`g'`y'5 = norm_enem_w _cons
matrix colnames b`x'`g'`y'6 = resid norm_enem_w _cons

}

foreach x in 1 2 4 5 {
forvalues j = 1/2 {
matrix bpval`g'`y'`x'[1,`j']=2*(normal(-abs(b`g'`y'`x'[1,`j']/bse`g'`y'`x'[1,`j'])))
matrix bpvalue`g'`y'`x'[1,`j']=(2 * ttail(DF`g'`y'`x', abs(b`g'`y'`x'[1,`j']/bse`g'`y'`x'[1,`j'])))
}
}

foreach x in 3 6 {
forvalues j = 1/3 {
matrix bpval`g'`y'`x'[1,`j']=2*(normal(-abs(b`g'`y'`x'[1,`j']/bse`g'`y'`x'[1,`j'])))
matrix bpvalue`g'`y'`x'[1,`j']=(2 * ttail(DF`g'`y'`x', abs(b`g'`y'`x'[1,`j']/bse`g'`y'`x'[1,`j'])))
}
}
forvalues i = 1(1)6 {
matrix list b`g'`y'`i'
matrix list bse`g'`y'`i'
matrix list bpval`g'`y'`i'
matrix list bpvalue`g'`y'`i'
}

* Adding my p values to the estimation results.
forvalues i=1(1)6 {
estimates restore `g'`y'`i'
estadd matrix pboot = bpval`g'`y'`i'
}

}


restore
}

* Average wages between 7-12 years after admission exam

esttab Fl_avg_annual_wage_712*  using "Output\Average_annual_wages_res_diff_ENEM_712_BS_Gender.tex", se star(* 0.10 ** 0.05 *** 0.01) drop(_cons) nomtitle f ///
stats(N ymean year program, fmt( %9.0fc %7.3f %3s %3s) labels("Number of observations"  "Mean dependent variable" "Exam year FE" "Major FE")) cells(b(star pvalue(pboot) fmt(%9.3f)) seboot(par fmt(%9.3f))) replace collabels(none) nogaps booktabs label /// 
refcat(resid " \\ \multicolumn{7}{l}{\textit{Panel A: Women}} \\", nolabel)

esttab Ml_avg_annual_wage_712*  using "Output\Average_annual_wages_res_diff_ENEM_712_BS_Gender.tex", se star(* 0.10 ** 0.05 *** 0.01) drop(_cons) nomtitle f ///
stats(N ymean year program , fmt( %9.0fc %7.3f %3s %3s) labels("Number of observations"  "Mean dependent variable" "Exam year FE" "Major FE")) cells(b(star pvalue(pboot) fmt(%9.3f)) seboot(par fmt(%9.3f))) append collabels(none) nogaps booktabs nonumber label ///
refcat(resid " \\ \multicolumn{7}{l}{\textit{Panel B: Men}} \\", nolabel)


